// File: @openzeppelin/contracts/access/AccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (access/AccessControl.sol)

pragma solidity ^0.8.0;

import "./IAccessControl.sol";
import "../utils/Context.sol";
import "../utils/Strings.sol";
import "../utils/introspection/ERC165.sol";

/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms. This is a lightweight version that doesn't allow enumerating role
 * members except through off-chain means by accessing the contract event logs. Some
 * applications may benefit from on-chain enumerability, for those cases see
 * {AccessControlEnumerable}.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Modifier that checks that an account has a specific role. Reverts
     * with a standardized message including the required role.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     *
     * _Available since v4.1._
     */
    modifier onlyRole(bytes32 role) {
        _checkRole(role);
        _;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    /**
     * @dev Revert with a standard message if `_msgSender()` is missing `role`.
     * Overriding this function changes the behavior of the {onlyRole} modifier.
     *
     * Format of the revert message is described in {_checkRole}.
     *
     * _Available since v4.6._
     */
    function _checkRole(bytes32 role) internal view virtual {
        _checkRole(role, _msgSender());
    }

    /**
     * @dev Revert with a standard message if `account` is missing `role`.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     */
    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(account),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view virtual override returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleGranted} event.
     */
    function grantRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleRevoked} event.
     */
    function revokeRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been revoked `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     *
     * May emit a {RoleRevoked} event.
     */
    function renounceRole(bytes32 role, address account) public virtual override {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * May emit a {RoleGranted} event.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     *
     * NOTE: This function is deprecated in favor of {_grantRole}.
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        bytes32 previousAdminRole = getRoleAdmin(role);
        _roles[role].adminRole = adminRole;
        emit RoleAdminChanged(role, previousAdminRole, adminRole);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleGranted} event.
     */
    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleRevoked} event.
     */
    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}

// File: @openzeppelin/contracts/access/IAccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator
    ) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1);

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator,
        Rounding rounding
    ) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10**64) {
                value /= 10**64;
                result += 64;
            }
            if (value >= 10**32) {
                value /= 10**32;
                result += 32;
            }
            if (value >= 10**16) {
                value /= 10**16;
                result += 16;
            }
            if (value >= 10**8) {
                value /= 10**8;
                result += 8;
            }
            if (value >= 10**4) {
                value /= 10**4;
                result += 4;
            }
            if (value >= 10**2) {
                value /= 10**2;
                result += 2;
            }
            if (value >= 10**1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10**result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result * 8) < value ? 1 : 0);
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

import "./math/Math.sol";

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: contracts/staking.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import '@openzeppelin/contracts/token/ERC721/IERC721.sol';
import '@openzeppelin/contracts/token/ERC721/IERC721Receiver.sol';
import '@openzeppelin/contracts/access/AccessControl.sol';

contract staking is IERC721Receiver, AccessControl {

    struct nftInformation {
        bool nftAccept;
        uint256 valueCoefficient;
    }
    mapping(address => nftInformation) public nftInformations;
    mapping(uint256 => address) public nftInformationAddresses;
    uint8 public nftInformationCounter;
    uint256 maxCoefficientValue;

    struct commercialType {
        bool commercialAccept;
        uint256 requirementValue;
    }
    mapping(uint8 => commercialType) public commercialTypes;
    uint8 public commercialTypeCounter;

    struct commercialUseRequest {
        bool isExist;
        uint8 commercialType;
        address requestAddress;
        bool useAccept;
        uint256 totalDepositValue;
        address[] coreAddress;
        uint256[] coreAmount;
        uint256[] coreIds;
        bool inUse;
        bool finishRequest;
        bool isFinished;
    }
    mapping(uint256 => commercialUseRequest) public commercialUseRequests;

    uint256 public commercialUseRequestCounter;

    event requestCommercialUseEvent(address holderAddress, uint256 commercialRequestNumber);

    struct userInformationTable {
        bool isExist;
        uint256 totalCoefficientValue;
        uint256[] commercialRequest;
        uint256[] commercialInUse;
        bool isMoved;
    }
    mapping(address => userInformationTable) public userInformationTables;

    struct userDepositToken {
        bool isExist;
        bool isDeposited;
        uint256[] coreProjects;
        bool isMoved;
    }
    mapping(address => mapping(address => mapping(uint256 => userDepositToken))) public userDepositTokens;

    mapping(address => mapping(address => uint256[])) public userDepositTokenIndex;


    mapping(address => bool) public stopWithdraw;
    mapping(address => address) public changeAddressEnable;
    
    constructor(){
        _setupRole(DEFAULT_ADMIN_ROLE, msg.sender);
    }

    function addNFTInformation(address _nftAddress, uint8 _coefficient) public onlyRole(DEFAULT_ADMIN_ROLE) {
        require(nftInformations[_nftAddress].nftAccept != true);
        require(_coefficient > 0, "need Coefficient");
        require(ERC165(_nftAddress).supportsInterface(type(IERC721).interfaceId) == true, "not ERC721");
        nftInformations[_nftAddress] = nftInformation(true, _coefficient);
        nftInformationAddresses[nftInformationCounter++] = _nftAddress;
        if (_coefficient > maxCoefficientValue) {
            maxCoefficientValue = _coefficient;
        }
    }

    function changeNftInformation(address _nftAddress, uint8 _coefficient) public onlyRole(DEFAULT_ADMIN_ROLE) {
        require(nftInformations[_nftAddress].nftAccept == true);
        require(_coefficient > 0, "need Coefficient");
        if (maxCoefficientValue == nftInformations[_nftAddress].valueCoefficient) {
            uint256 _index = _coefficient;
            for (uint i = 0; i < nftInformationCounter; i++) {
                address _searchNftAddress = nftInformationAddresses[i];
                if (_searchNftAddress == _nftAddress) {
                    // skip
                } else if (nftInformations[_nftAddress].valueCoefficient > _index) {
                    _index = nftInformations[_nftAddress].valueCoefficient;
                }
            }
            maxCoefficientValue = _index;
        }
        nftInformations[_nftAddress] = nftInformation(true, _coefficient);
    }

    function stopNFTInformation(address _nftAddress) public onlyRole(DEFAULT_ADMIN_ROLE) {
        require(nftInformations[_nftAddress].nftAccept == true);
        nftInformations[_nftAddress].nftAccept = false;
    }

    function addCommercialType(uint256 _requirementValue) public onlyRole(DEFAULT_ADMIN_ROLE) {
        require(_requirementValue > 0);
        commercialTypes[commercialTypeCounter++] = commercialType(true, _requirementValue);
    }

    function changeCommercialType(uint8 _commercialNumber, uint256 _requirementValue) public onlyRole(DEFAULT_ADMIN_ROLE) {
        require(_requirementValue > 0);
        commercialTypes[_commercialNumber] = commercialType(true, _requirementValue);
    }

    function stopCommercialType(uint8 _commercialNumber) public onlyRole(DEFAULT_ADMIN_ROLE) {
        commercialTypes[_commercialNumber] = commercialType(false, 0);
    }

    function checkUserInformationTable(address _holder) public view returns(userInformationTable memory) {
        return userInformationTables[_holder];
    }

    function checkUserDepositToken(address _holder, address _nftAddress, uint256 _nftId) public view returns(userDepositToken memory) {
        return userDepositTokens[_holder][_nftAddress][_nftId];
    }

    function checkUserDepositTokenIndex(address _holder, address _nftAddress) public view returns(uint256[] memory) {
        return userDepositTokenIndex[_holder][_nftAddress];
    }

    function checkCommercialUseEnable(address _holder, uint8 _requestCommercialType) public view returns(bool) {
        require(_requestCommercialType < commercialTypeCounter);
        uint256 _totalNftDepositValues;

        if (userInformationTables[_holder].isExist == true) {
            _totalNftDepositValues += userInformationTables[_holder].totalCoefficientValue;
        }

        for (uint i = 0; i < nftInformationCounter; i++) {
            address _nftAddress = nftInformationAddresses[i];
            
            _totalNftDepositValues += IERC721(_nftAddress).balanceOf(_holder) * nftInformations[_nftAddress].valueCoefficient;

        }
        uint256 _requirementNftValues = commercialTypes[_requestCommercialType].requirementValue;
        if(_totalNftDepositValues >= _requirementNftValues) {
            return true;
        } else {
            return false;
        }
    }

    function requestCommercialUse(
        uint8 _requestCommercialType,
        address[] memory _coreNftAddress,
        uint256[] memory _coreNftAmount,
        uint256[] memory _coreNftIds
    ) public {
        require(commercialTypes[_requestCommercialType].commercialAccept == true);
        require(checkCommercialUseEnable(msg.sender, _requestCommercialType) == true);
        if (_coreNftAddress.length >= 1) {
            require(_coreNftAddress.length == _coreNftAmount.length);
            uint256 _amountRepeatMarker;

            for (uint256 i = 0; i < _coreNftAddress.length; i++) {
                address _nftAddress = _coreNftAddress[i];
                for (uint256 j = _amountRepeatMarker; j < _amountRepeatMarker + _coreNftAmount[i]; j++) {
                    uint256 _nftId = _coreNftIds[j];

                    if (IERC721(_nftAddress).ownerOf(_nftId) == msg.sender) {}
                    else if (IERC721(_nftAddress).ownerOf(_nftId) == address(this)) {
                        require(userDepositTokens[msg.sender][_nftAddress][_nftId].isDeposited == true, "Tokens deposited by others");
                    } else {
                        revert("You are not owner");
                    }

                }
                _amountRepeatMarker += _coreNftAmount[i];
            }

            commercialUseRequests[commercialUseRequestCounter].coreAddress = _coreNftAddress;
            commercialUseRequests[commercialUseRequestCounter].coreAmount = _coreNftAmount;
            commercialUseRequests[commercialUseRequestCounter].coreIds = _coreNftIds;
        }
        commercialUseRequests[commercialUseRequestCounter].isExist = true;
        commercialUseRequests[commercialUseRequestCounter].commercialType = _requestCommercialType;
        commercialUseRequests[commercialUseRequestCounter].requestAddress = msg.sender;

        if (userInformationTables[msg.sender].isExist != true) {
            userInformationTables[msg.sender].isExist = true;
        }

        userInformationTables[msg.sender].commercialRequest.push(commercialUseRequestCounter);
        emit requestCommercialUseEvent(msg.sender, commercialUseRequestCounter);
        commercialUseRequestCounter++;
    }

    function acceptCommercialUse(uint256 _commercialRequestNumber) public onlyRole(DEFAULT_ADMIN_ROLE) {
        require(commercialUseRequests[_commercialRequestNumber].isExist == true, "not exist");
        require(commercialUseRequests[_commercialRequestNumber].useAccept != true, "already accepted");
        commercialUseRequests[_commercialRequestNumber].useAccept = true;
    }

    function startCommercialUse(
        uint256 _commercialRequestNumber,
        address[] memory _address,
        uint256[] memory _depositAmount,
        uint256[] memory _depositIds
    ) public {
        require(commercialUseRequests[_commercialRequestNumber].useAccept == true, "not enable");
        require(commercialUseRequests[_commercialRequestNumber].requestAddress == msg.sender, "you are not request owner");

        uint256 _alreadyDepositedCoefficientValue = userInformationTables[msg.sender].totalCoefficientValue;
        uint8 _requestCommercialType = commercialUseRequests[_commercialRequestNumber].commercialType;
        uint256 _requirementCoefficientValue = commercialTypes[_requestCommercialType].requirementValue;

        require(_address.length == _depositAmount.length);
        uint256 _totalCoefficient = _alreadyDepositedCoefficientValue;
        uint256 _amountRepeatMarker;

        for (uint256 i = 0; i < _address.length; i++) {
            address _nftAddress = _address[i];
            require(nftInformations[_nftAddress].nftAccept == true, "NFT is not enabled");
            for (uint256 j = _amountRepeatMarker; j < _amountRepeatMarker + _depositAmount[i]; j++) {
                uint256 _nftId = _depositIds[j];

                IERC721(_nftAddress).safeTransferFrom(msg.sender, address(this), _nftId);

                userDepositTokens[msg.sender][_nftAddress][_nftId].isExist = true;
                userDepositTokens[msg.sender][_nftAddress][_nftId].isDeposited = true;
                userDepositTokenIndex[msg.sender][_nftAddress].push(_nftId);
            }
            _totalCoefficient += _depositAmount[i] * nftInformations[_nftAddress].valueCoefficient;
            _amountRepeatMarker += _depositAmount[i];
        }
        require(_totalCoefficient >= _requirementCoefficientValue, "not enough NFTs");
        require(_amountRepeatMarker == _depositIds.length, "deposit amount is not equal ids");
        userInformationTables[msg.sender].totalCoefficientValue = _totalCoefficient;

        if (commercialUseRequests[_commercialRequestNumber].coreAddress.length >= 1) {
            uint256 _coreAmountRepeatMarker;
            for (uint256 i = 0; i < commercialUseRequests[_commercialRequestNumber].coreAddress.length; i++) {
                address _coreNftAddress = commercialUseRequests[_commercialRequestNumber].coreAddress[i];
                for (uint256 j = _coreAmountRepeatMarker; j < _coreAmountRepeatMarker + commercialUseRequests[_commercialRequestNumber].coreAmount[i]; j++) {
                    uint256 _coreNftId = commercialUseRequests[_commercialRequestNumber].coreIds[j];
                    require(userDepositTokens[msg.sender][_coreNftAddress][_coreNftId].isDeposited == true, "core NFT does not transfered");
                    userDepositTokens[msg.sender][_coreNftAddress][_coreNftId].coreProjects.push(_commercialRequestNumber);
                }
                _coreAmountRepeatMarker += commercialUseRequests[_commercialRequestNumber].coreAmount[i];
            }
        }

        commercialUseRequests[_commercialRequestNumber].inUse = true;

        userInformationTables[msg.sender].commercialInUse.push(_commercialRequestNumber);
    }

    function finishCommercialRequest(uint256 _commercialRequestNumber) public {
        require(commercialUseRequests[_commercialRequestNumber].requestAddress == msg.sender, "you are not request owner");
        require(commercialUseRequests[_commercialRequestNumber].finishRequest != true, "already send finish request");
        commercialUseRequests[_commercialRequestNumber].finishRequest = true;
    }

    function acceptFinishRequest(uint256 _commercialRequestNumber) public onlyRole(DEFAULT_ADMIN_ROLE) {
        require(commercialUseRequests[_commercialRequestNumber].finishRequest == true, "already send finish request");
        commercialUseRequests[_commercialRequestNumber].isFinished = true;

        address _holderAddress = commercialUseRequests[_commercialRequestNumber].requestAddress;

        if (commercialUseRequests[_commercialRequestNumber].coreAddress.length != 0) {
            uint256 _amountRepeatMarker;
            
            for (uint256 i = 0; i < commercialUseRequests[_commercialRequestNumber].coreAddress.length; i++) {
                address _nftAddress = commercialUseRequests[_commercialRequestNumber].coreAddress[i];
                for (uint j = _amountRepeatMarker; j < _amountRepeatMarker + commercialUseRequests[_commercialRequestNumber].coreAmount[i]; j++) {
                    uint256 _nftId = commercialUseRequests[_commercialRequestNumber].coreIds[j];
                    uint256 _index;
                    
                    for (uint256 k = 0; k < userDepositTokens[_holderAddress][_nftAddress][_nftId].coreProjects.length; k++) {
                        if(userDepositTokens[_holderAddress][_nftAddress][_nftId].coreProjects[k] == _commercialRequestNumber) {
                            _index = k;
                            break;
                        }
                    }

                    userDepositTokens[_holderAddress][_nftAddress][_nftId].coreProjects[_index] = userDepositTokens[_holderAddress][_nftAddress][_nftId].coreProjects[userDepositTokens[_holderAddress][_nftAddress][_nftId].coreProjects.length - 1];

                    userDepositTokens[_holderAddress][_nftAddress][_nftId].coreProjects.pop();
                }
                _amountRepeatMarker += commercialUseRequests[_commercialRequestNumber].coreAmount[i];
            }
        }

        if (userInformationTables[_holderAddress].commercialInUse.length == 1) {
            userInformationTables[_holderAddress].commercialInUse.pop();
        } else {
            uint256 _userInformationTableIndex;
            for (uint256 i = 0; i < userInformationTables[msg.sender].commercialInUse.length; i++) {
                if (userInformationTables[msg.sender].commercialInUse[i] == _commercialRequestNumber) {
                    _userInformationTableIndex = i;
                    break;
                }
            }
            userInformationTables[_holderAddress].commercialInUse[_userInformationTableIndex] = userInformationTables[_holderAddress].commercialInUse[userInformationTables[_holderAddress].commercialInUse.length - 1];
        }
    }

    function depositNft(address _nftAddress, uint256 _nftId) public {
        require(nftInformations[_nftAddress].nftAccept == true, "NFT is not enabled");
        IERC721(_nftAddress).safeTransferFrom(msg.sender, address(this), _nftId);
        userDepositTokens[msg.sender][_nftAddress][_nftId].isExist = true;
        userDepositTokens[msg.sender][_nftAddress][_nftId].isDeposited = true;
        userDepositTokenIndex[msg.sender][_nftAddress].push(_nftId);
        userInformationTables[msg.sender].totalCoefficientValue += nftInformations[_nftAddress].valueCoefficient;
    }

    function batchDepositNft(address[] memory _nftAddresses, uint256[] memory _nftAmounts, uint256[] memory _nftIds) public {
        require(_nftAddresses.length == _nftAmounts.length);
        uint256 _depositAmountMarker;
        for (uint256 i = 0; i < _nftAddresses.length; i++) {
            address _nftAddress = _nftAddresses[i];
            for (uint j = _depositAmountMarker; j < _depositAmountMarker + _nftAmounts[i]; j++) {
                uint256 _nftId = _nftIds[j];
                depositNft(_nftAddress, _nftId);
            }
            _depositAmountMarker += _nftAmounts[i];
        }
        require(_depositAmountMarker == _nftIds.length, "Amount and Ids are not same");
    }

    function withDraw(address _nftAddress, uint256 _nftId) public {
        require(stopWithdraw[msg.sender] == false);
        require(userDepositTokens[msg.sender][_nftAddress][_nftId].isDeposited == true, "not deposited");
        require(userDepositTokens[msg.sender][_nftAddress][_nftId].coreProjects.length == 0, "token has core project");
        
        require(nftInformations[_nftAddress].nftAccept == true, "unenabled address");

        uint256 _userCoefficientValue = userInformationTables[msg.sender].totalCoefficientValue;
        uint256 _tokenCoefficient = nftInformations[_nftAddress].valueCoefficient;
        uint256 _resultCoefficientValue = _userCoefficientValue - _tokenCoefficient;

        if (_resultCoefficientValue >= maxCoefficientValue) {
            //skip
        } else if (userInformationTables[msg.sender].commercialInUse.length == 0) {
            //skip
        } else {
            for (uint i = 0; i < userInformationTables[msg.sender].commercialInUse.length; i++) {
                uint256 _commercialRequestNumber = userInformationTables[msg.sender].commercialInUse[i];
                uint8 _commercialTypeNumber = commercialUseRequests[_commercialRequestNumber].commercialType;
                uint256 _requirementCoefficient = commercialTypes[_commercialTypeNumber].requirementValue;

                require(_resultCoefficientValue >= _requirementCoefficient);

            }
        }

        IERC721(_nftAddress).safeTransferFrom(address(this), msg.sender, _nftId);

        userDepositTokens[msg.sender][_nftAddress][_nftId].isDeposited = false;

        uint256 _userDepositTokenAmount = userDepositTokenIndex[msg.sender][_nftAddress].length;
        uint256 _userDepositTokenIndex;
        for (uint i = 0; i < _userDepositTokenAmount; i++) {
            if (_nftId == userDepositTokenIndex[msg.sender][_nftAddress][i]) {
                _userDepositTokenIndex = i;
                break;
            }
        }
        userDepositTokenIndex[msg.sender][_nftAddress][_userDepositTokenIndex] = userDepositTokenIndex[msg.sender][_nftAddress][userDepositTokenIndex[msg.sender][_nftAddress].length -1];
        userDepositTokenIndex[msg.sender][_nftAddress].pop();

        uint256 _coefficient = nftInformations[_nftAddress].valueCoefficient;
        userInformationTables[msg.sender].totalCoefficientValue -= _coefficient;

    }

    function batchWithDraw(address[] memory _nftAddresses, uint256[] memory _nftAmounts, uint256[] memory _nftIds) public {
        require(_nftAddresses.length == _nftAmounts.length);
        uint256 _depositAmountMarker;
        for (uint256 i = 0; i < _nftAddresses.length; i++) {
            address _nftAddress = _nftAddresses[i];
            for (uint j = _depositAmountMarker; j < _depositAmountMarker + _nftAmounts[i]; j++) {
                uint256 _nftId = _nftIds[j];
                withDraw(_nftAddress, _nftId);
            }
            _depositAmountMarker += _nftAmounts[i];
        }
        require(_depositAmountMarker == _nftIds.length, "Amount and Ids are not same");
    }

    function checkCommercialUseRequest(uint8 _requestCommercialType) public view returns(commercialUseRequest memory) {
        return commercialUseRequests[_requestCommercialType];
    }

    function setStopWithdraw(address _holderAddress) public onlyRole(DEFAULT_ADMIN_ROLE) {
        require(stopWithdraw[_holderAddress] == false);
        stopWithdraw[_holderAddress] = true;
    }

    function restartWithdraw(address _holderAddress) public onlyRole(DEFAULT_ADMIN_ROLE) {
        require(stopWithdraw[_holderAddress] == true);
        stopWithdraw[_holderAddress] = false;
    }

    function setChangeAddressEnable(address _oldAddress, address _newAddress) public onlyRole(DEFAULT_ADMIN_ROLE) {
        require(userInformationTables[_newAddress].isExist != true);
        changeAddressEnable[_oldAddress] = _newAddress;
        if(stopWithdraw[_oldAddress] != true) {
            stopWithdraw[_oldAddress] = true;
        }
    }

    function changeAddress(address _oldAddress) public {
        require(changeAddressEnable[_oldAddress] == msg.sender);
        require(userInformationTables[msg.sender].isExist != true);

        userInformationTables[msg.sender] = userInformationTables[_oldAddress];
        userInformationTables[_oldAddress].isMoved = true;

        for (uint i = 0; i < nftInformationCounter; i++) {
            address _nftAddress = nftInformationAddresses[i];
            uint256 _nftAmount = userDepositTokenIndex[_oldAddress][_nftAddress].length;

            if (_nftAmount >= 1) {
                userDepositTokenIndex[msg.sender][_nftAddress] = userDepositTokenIndex[_oldAddress][_nftAddress];
                for (uint j = 0; j < _nftAmount; j++) {
                    uint256 _nftId = userDepositTokenIndex[_oldAddress][_nftAddress][j];
                    userDepositTokens[msg.sender][_nftAddress][_nftId] = userDepositTokens[_oldAddress][_nftAddress][_nftId];
                    userDepositTokens[_oldAddress][_nftAddress][_nftId].isMoved = true;
                    userDepositTokens[_oldAddress][_nftAddress][_nftId].isDeposited = true;
                }
            }
        }

        uint256 _commercialInformationAmount = userInformationTables[msg.sender].commercialRequest.length;
        for (uint i = 0; i < _commercialInformationAmount; i++) {
            uint256 _commercialInformationNumber = userInformationTables[msg.sender].commercialRequest[i];
            commercialUseRequests[_commercialInformationNumber].requestAddress = msg.sender;
        }
    }


    function onERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public override returns(bytes4) {
        require(nftInformations[msg.sender].nftAccept == true);
        return this.onERC721Received.selector;
    }

}